

AI Search LOG

Zhuo Chen, Martha Cao, Ted Lystig, and Sateesh Arjula
BridgeBio Pharma, LLC

ABSTRACT

In a clinical study, it is important to make sure LOG files generated by SAS programs are clean. Search LOG files maybe needed daily during the study program development cycle for SDTM, ADaM, Tables Listings and Figures etc.

This paper describes an automated method to reduce human labor in searching LOG efforts therefore helps team to gain efficiency.


INTRODUCTION

This paper presents one simple solution to automate LOG check. After initial set up, it will automatically run SearchLOG batch file and macro every day at team preferred time. This macro will search all LOG files under any specified study folder for issues. Then it will automatically email SAS users whose LOG files contain issues. It will also automatically generate an EXCEL file with all issues, associated LOG file name, generation datetime, owner and owners email address to function as a convenient tracker for programming team.

STEP 1: AUTOMATE SEARCH LOG BATCH RUN


First, Windows Task Scheduler is used to set up an automative run of SAS batch file. The example below shows major steps to set up one batch file run every midnight. Your team can choose your preferred frequency and datetime.

Display 1 shows how to Create a Basic Task via Windows Task Scheduler and add Name and Description for this task.



Display 1. Create a Basic Search LOG Task


Display 2 shows different options for task run frequency or start conditions. This paper chooses Daily run as an example. You can choose different options based on your needs.



The screenshot shows the 'Create Basic Task Wizard' window with the title 'Task Trigger'. On the left, there is a sidebar with 'Trigger' selected, and 'Action' and 'Finish' are visible below it. The main area is titled 'When do you want the task to start?' and contains seven radio button options: 'Daily' (selected), 'Weekly', 'Monthly', 'One time', 'When the computer starts', 'When I log on', and 'When a specific event is logged'. At the bottom right, there are three buttons: '< Back', 'Next >', and 'Cancel'.

Display 2. Setup Task Start Frequency or Condition


Display 3 shows how to set up the start date, time and frequency. Example in this paper chooses run Search LOG every midnight.



The screenshot shows the 'Create Basic Task Wizard' window with the title 'Daily'. On the left, there is a sidebar with 'Daily' selected, and 'Trigger' and 'Finish' are visible above it. The main area contains a 'Start' section with a date field set to '1/15/2025', a time field set to '12:00:00 AM', and a checkbox for 'Synchronize across time zones'. Below this is a 'Recur every:' section with a text box containing '1' and the word 'days'. At the bottom right, there are three buttons: '< Back', 'Next >', and 'Cancel'.


Display 3. Setup Task Start Date, Time and Frequency

Display 4 shows the Action for this Task Scheduler. In this example, since the goal is to autorun a SAS batch file every night, “Start a program” has been chosen.




Display 4. Setup Action

Display 5 shows how to specify the location and SAS batch file. Since most study set up SDTM, ADaM, Tables, Listings and Figures on different folders, you can create different batch files, each batch file for each folder. Another option is to set up SearchLOG for different folders in one batch file. It all depends on team's needs.



Display 5. Specify Batch File

Display 6 demonstrates a summary for what has been set up in Task Scheduler. In this example, it is autorun a SAS batch file at every midnight.



Display 6. Summary of a Scheduled Task as Example

STEP 2: CREATE BATCH FILE

Below is an example of batch file created to run under SAS Enterprise Guide 8.4 environment. After finishing set up in STEP 1, this batch file will automatically run macro AISerachLog under folder G:\CompoundA\StudyB\DeliverableC\tables\programs using SAS version 9.4 every midnight. After each run, an AISearchLog.log file will be created under the same folder to help team debug if needed.

```
set sas=E:\SASHome\SASFoundation\9.4\sas.exe
set proj=G:\CompoundA\StudyB\DeliverableC\tables\programs
set log= G:\CompoundA\StudyB\DeliverableC\tables\programs
set name1=AIsearchLog
set pgm1=%proj%\%name1%.sas
set log1=%log%\%name1%.log
%sas% -sysin %pgm1% -log %log1% quit;
```

Program 1. Example Batch File SearchLog.bat

Sometimes team want to refresh all outputs first then run SearchLOG. Below is another example to accomplish that.

```
set sas=E:\SASHome\SASFoundation\9.4\sas.exe
set proj=G:\CompoundA\StudyB\DeliverableC\tables\programs
set log= G:\CompoundA\StudyB\DeliverableC\tables\programs
set name1=t-1-1
set name2=t-1-2
set name3=AIsearchLog
set pgm1=%proj%\%name1%.sas
set pgm2=%proj%\%name2%.sas
set pgm3=%proj%\%name3%.sas
```

```

set log1=%log%\%name1%.log
set log2=%log%\%name2%.log
set log3=%log%\%name3%.log
%sas% -sysin %pgm1% -log %log1% quit;
%sas% -sysin %pgm2% -log %log2% quit;
%sas% -sysin %pgm3% -log %log3% quit;

```

Program 2. Another example of Batch File to run multiple SAS programs first, then run SearchLog

STEP 3: MACRO %AISEARCHLOG

This macro has only one macro variable: &loc which is the folder path of LOG files.

After passing the folder path macro variable, this macro will automatically find all LOG files under that folder and check LOG files one by one based on key words your team identified. Then it will output all issues and associated LOG file information into an EXCEL file under that same folder. Please see one sample out in Output 1 below:

	A	B	C	D	E
1	LOG File Name	Message in LOG File	LOG File Generation Datetime	LOG File Generated By	Email Address
2	f-1-1.Log	something invalid	08/21/2024 01:15PM	BRIDGEBIOSAS\jchen	joyce.chen@bridgebio.com
3	f-1-2.Log	something invalid	08/21/2024 01:15PM	BRIDGEBIOSAS\jchen	joyce.chen@bridgebio.com
4	l-1-1-7.log	something invalid	08/21/2024 01:15PM	BRIDGEBIOSAS\jchen	joyce.chen@bridgebio.com
5	l-1-1.log	something invalid	08/21/2024 01:15PM	BRIDGEBIOSAS\jchen	joyce.chen@bridgebio.com
6	t-1-1.LOG	ERROR	01/09/2025 09:35AM	BRIDGEBIOSAS\jchen	joyce.chen@bridgebio.com
7	t-1-1.LOG	WARNING	01/09/2025 09:35AM	BRIDGEBIOSAS\jchen	joyce.chen@bridgebio.com
8	t-1-1.LOG	UnInitialized	01/09/2025 09:35AM	BRIDGEBIOSAS\jchen	joyce.chen@bridgebio.com
9	t-1-2.log	something Invalid	01/09/2025 09:33AM	BRIDGEBIOSAS\jchen	joyce.chen@bridgebio.com
10	t-1-3.Log	something invalid	08/21/2024 01:15PM	BRIDGEBIOSAS\jchen	joyce.chen@bridgebio.com
11	v-t-1-1.log	WARNING: Unable to copy SASUSER registry to WORK re	03/20/2023 10:37AM	BRIDGEBIOSAS\jchen	joyce.chen@bridgebio.com
12	v-t-1-2.log	WARNING: Unable to copy SASUSER registry to WORK re	03/20/2023 10:37AM	BRIDGEBIOSAS\jchen	joyce.chen@bridgebio.com

Output 1. Sample of AISearchLogFindings.XLSX

As you can see, this macro not only lists LOG file which has issues and what's the issue, it also listed LOG file creation datetime and creator's name.

Since each SAS user has his/her unique email address, this macro uses SAS user ID to find the corresponding email address of that SAS user, then call the sub macro %email by passing that users email address and send an email to that SAS user. Therefore, only users whose LOG has issues receive alert email automatically. Program 3 below contains SAS code and explanations of each section of this macro.

```

%macro AISearchLog(loc=);
options mlogic mprint;

* find all LOG files under folder &loc, and associated creator and creation datetime;
filename nams pipe "dir /Q &loc";

data file ;
  infile nams missover pad length=len;
  input @01 all $varying200. len ;
  if index(upcase(all),'.LOG');
  dt=scan(trim(left(all)),1," ")||" "||scan(trim(left(all)),2,"
")||scan(trim(left(all)),3," ");
  user=scan(trim(left(all)),5," ");
  log=scan(trim(left(all)),6," ");
run;

data mytemp;
  tot=0;
  retain tot;
  set file end=eof;
  tot=_n_;
  j=put(tot,best.);

```

```

        call symput("f"||left(j),log);
        if eof then call symput("tot",tot);
run;

data logsum;
    length dsname $200. message $2000. ;
    dsname=""; message="";
run;

* loop through all LOG files under &loc, only keep LOG file with issues in dataset
logsum;
%do i=1 %to &tot;
    filename onelog "&loc\&&f&i";
    data log&i;
        infile onelog pad length=len missover end=eof;
        input @01 message $varying200. len;
        if index(upcase(message), "ERROR") then flag=1;
        if index(upcase(message), "WARNING") then flag=1;
        if index(upcase(message), "UNINITIALIZED") then flag=1;
        if index(upcase(message), "INVALID") then flag=1;
        /* you can add more keywords based on your teams decision */
        if flag=1;
        dsname="&&f&i";
    run;
    data logsum (keep=message dsname);
        set logsum log&i;
        if message='' and dsname='' then delete;
    run;
%end;

* use email.xlsx file to find email address of each LOG creator;
proc import out=email
    datafile="G:\global\email.xlsx"
    dbms=xlsx replace;
    getnames=YES;
run;

proc sort data=email; by user; run;
proc sort data=file; by user; run;
data file;
    length dsname $200.;
    merge file (in=x) email;
        by user;
    if x;
        dsname=log;
run;
proc sort data=file; by dsname; run;

* create final data set which contains all LOG files with issues;
data final (keep=message dsname user dt email);
    merge logsum (in=x) file (in=y);
        by dsname;
    if x;
        label dsname="LOG File Name"
            message="Message in LOG File"
            dt="LOG File Generation Datetime"
            user="LOG File Generated By"
            email="Email Address";
run;

* output it into AISearchLogFindings.xlsx under the same folder &loc;
proc export data=final
    outfile="&loc\AISearchLogFindings.xlsx"
    dbms=xlsx label
    replace;
    sheet="&loc";
run;

```

```

* send emails to LOG creators whose LOG has issue;
proc sort data=final (keep=email) out=sendemail nodupkey;
  by email;
run;

data mytemp2;
  tot2=0;
  retain tot2;
  set sendemail end=eof;
  tot2=_n_;
  j=put(tot2,best.);
  call symput("email"||left(j),email);
  if eof then call symput("tot2",tot2);
run;

%do i=1 %to &tot2;
  %email(to=&&email&i,loc=&loc);
%end;
%mend AISearchLog;

*example;
%AISearchLog(loc=G:\CompoundA\StudyB\DeliverableC\tables\programs);

```

Program 3. Macro %AISearchLog

Please note that this macro requires no space in the path (&loc) and the length of SAS User ID <=22.

Program 4 below is the %email macro which used SMTP method. Please note that both the id and password are dummy in example below. You may need help from your SAS system manager or IT to set up SMTP so it enables your SAS system to send email to SAS users.

```

options emailsys=smtp emailhost=
(
  "smtp.gmail.com"
  port=587 STARTTLS
  auth=login
  id="userdummy558@gmail.com"
  pw="{SAS005}DUMMY"
)
;

%macro email(to=,loc=);
filename AEmail EMAIL DEBUG
  to=("&to.")
  Subject="AI Search LOG Findings";


data _null_;
  file AEmail;
  put "Please check your LOG file under &loc.";
  put "(Refer to AISearchLogFindings.xlsx on the same folder for details).";
run;

filename AEmail clear;
quit;
%mend email;

```

Program 4 . Sub Macro %email

Output 2 below shows an example of an email received. Team can customize the contents in the email based on your needs.



AI Search LOG Findings

From userdummy558@gmail.com <userdummy558@gmail.com>

Date Sun 3/23/2025 12:06 AM

To Joyce Chen <joyce.chen@bridgebio.com>

Please check your LOG file under G:\CompoundA\StudyB\DeliverableC\tables\programs
(Refer to AISearchLogFindings.xlsx on the same folder for details).

Output 2 . Sample of Email Received Automatically

CONCLUSION

The method described in this paper combines the functionality of SAS, Windows Task Scheduler and SMPT Email System to provide an automatic search LOG solution. It only alerts owners whose LOG file has issues. Therefore, it is called AI Search LOG.

ACKNOWLEDGMENTS

Many thanks to Tuan Nguyen, Patrick Moriarty, Brian Tague and Jon Amundsen for setting up password and providing test code so SMPT system can support email in SAS.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Zhuo Chen (Joyce)
BridgeBio Pharma, LLC
Joyce.Chen@bridgebio.com

Martha Cao
BridgeBio Pharma, LLC
Martha.Cao@bridgebio.com

Ted Lystig
BridgeBio Pharma, LLC
Ted.Lystig@bridgebio.com

Sateesh Arjula
BridgeBio Pharma, LLC
Sateesh.Arjula@bridgebio.com